NAG Library Routine Document

F04AHF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F04AHF calculates the accurate solution of a set of real linear equations with multiple right-hand sides, $AX = B$, with iterative refinement, where A has been factorized by F03AFF.

2 Specification

```
SUBROUTINE F04AHF (N, IR, A, LDA, AA, LDAA, P, B, LDB, EPS, X, LDX, BB,        &
                   LDBB, K, IFAIL)
INTEGER            N, IR, LDA, LDAA, LDB, LDX, LDBB, K, IFAIL
REAL (KIND=nag_wp) A(LDA,N), AA(LDAA,N), P(N), B(LDB,IR), EPS, X(LDX,IR),      &
                   BB(LDBB,IR)
```

3 Description

To solve a set of real linear equations $AX = B$, F04AHF must be preceded by a call to F03AFF which computes an LU factorization of A with partial pivoting, $PA = LU$, where P is a permutation matrix, L is lower triangular and U is unit upper triangular. An approximation to X is found by forward and backward substitution. The residual matrix $R = B - AX$ is then calculated using ***additional precision***, and a correction D to X is found by solving $LUD = PR$. X is replaced by $X + D$, and this iterative refinement of the solution is repeated until full machine accuracy has been obtained.

4 References

Wilkinson J H and Reinsch C (1971) *Handbook for Automatic Computation II, Linear Algebra* Springer–Verlag

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 0.

2: IR – INTEGER *Input*

On entry: r, the number of right-hand sides.

3: A(LDA,N) – REAL (KIND=nag_wp) array *Input*

On entry: the n by n matrix A.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F04AHF is called.

Constraint: LDA $\geq$ N.

5: AA(LDAA,N) – REAL (KIND=nag_wp) array *Input*

On entry: details of the LU factorization, as returned by F03AFF.

6: LDAA – INTEGER *Input*

On entry: the first dimension of the array AA as declared in the (sub)program from which F04AHF is called.

Constraint: LDAA $\geq$ N.

7: P(N) – REAL (KIND=nag_wp) array *Input*

On entry: details of the row interchanges as returned by F03AFF.

8: B(LDB,IR) – REAL (KIND=nag_wp) array *Input*

On entry: the n by r right-hand side matrix B.

9: LDB – INTEGER *Input*

On entry: the first dimension of the array B as declared in the (sub)program from which F04AHF is called.

Constraint: LDB $\geq$ N.

10: EPS – REAL (KIND=nag_wp) *Input*

On entry: must be set to the value of the ***machine precision***.

11: X(LDX,IR) – REAL (KIND=nag_wp) array *Output*

On exit: the n by r solution matrix X.

12: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which F04AHF is called.

Constraint: LDX $\geq$ N.

13: BB(LDBB,IR) – REAL (KIND=nag_wp) array *Output*

On exit: the final n by r residual matrix $R = B - AX$.

14: LDBB – INTEGER *Input*

On entry: the first dimension of the array BB as declared in the (sub)program from which F04AHF is called.

Constraint: LDBB $\geq$ N.

15: K – INTEGER *Output*

On exit: the number of iterations needed in the refinement process.

16: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

The matrix A is too ill-conditioned to produce a correctly rounded solution.

7 Accuracy

The computed solutions should be correct to full machine accuracy. For a detailed error analysis see page 106 of Wilkinson and Reinsch (1971).

8 Further Comments

The time taken by F04AHF is approximately proportional to $n^2 r$.

9 Example

This example solves the set of linear equations $AX = B$ where

$$A = \begin{pmatrix} 33 & 16 & 72 \\ -24 & -10 & -57 \\ -8 & -4 & -17 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} -359 \\ 281 \\ 85 \end{pmatrix}.$$

9.1 Program Text

```
    Program f04ahfe

!     F04AHF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: f03aff, f04ahf, nag_wp, x02ajf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: d1, eps
      Integer                          :: i, id, ifail, ir, k, lda, ldaa, ldb, &
                                          ldbb, ldx, n
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), aa(:,:), b(:,:), bb(:,:),    &
                                          p(:), x(:,:)
!     .. Executable Statements ..
      Write (nout,*) 'F04AHF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      ir = 1
      lda = n
      ldaa = n
      ldb = n
      ldbb = n
      ldx = n
      Allocate (a(lda,n),aa(ldaa,n),b(ldb,ir),bb(ldbb,ir),p(n),x(ldx,ir))
      Read (nin,*)(aa(i,1:n),i=1,n)
      a(1:n,1:n) = aa(1:n,1:n)
```

```
!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
!     Crout decomposition
      Call f03aff(n,eps,aa,ldaa,d1,id,p,ifail)

      Read (nin,*)(b(i,1:ir),i=1,n)
      eps = x02ajf()

      ifail = 0
!     Accurate solution of linear equations
      Call f04ahf(n,ir,a,lda,aa,ldaa,p,b,ldb,eps,x,ldx,bb,ldbb,k,ifail)

      Write (nout,*) ' Solution'
      Do i = 1, n
        Write (nout,99999) x(i,1:ir)
      End Do

99999 Format (1X,8F9.4)
    End Program f04ahfe
```

9.2 Program Data

```
F04AHF Example Program Data
  3
   33   16   72
  -24  -10  -57
   -8   -4  -17
 -359  281   85
```

9.3 Program Results

```
 F04AHF Example Program Results

  Solution
    1.0000
   -2.0000
   -5.0000
```